```
> restart: with(DEtools): with(plots):
Warning, the name changecoords has been redefined
```

Procédures de développement limité

```
> dl2:=proc(X) global e; local i; if type(X,list) then
  combine([seq(convert(taylor(X[i],e,2),polynom),i=1..nops(X))]
  ,trig) else  taylor(X,e,2) end if; end proc:
> dleq:=proc(eq) dl2(lhs(eq))=dl2(rhs(eq)) end proc:
> dlsolve:=proc(eq,x,n) global e; local sol,X,EQ;
  EQ:=subs(theta(t)=sum(X[i](t)*e^i,i=0..n-1),eq);
  sol:=dsolve({seq(coeftayl(lhs(EQ),e=0,i)=coeftayl(rhs(EQ),e=0
  ,i),i=0..n-1),
  seq(X[i](0)=0,i=0..n-1)},[seq(X[i](t),i=0..n-1)]);
  combine(subs(sol,sum(X[i](t)*e^i,i=0..n-1)),trig); end proc:
> dltheta:=proc(coord,n) global e; local theta;
  theta:=arctan(coord[2],coord[1]);
  combine(convert(simplify(taylor(theta,e,n),symbolic),polynom)
  ,trig) end proc:
> dlcmp:=proc(X,Y,n) global e; local i;
  sum('(coeftayl(X,e=0,i)-coeftayl(Y,e=0,i))*e^i','i'=1..n);
  end proc:
> arcmax:=proc(err) global e,omega,t; local
  arc,alpha,s,s1,s2,m1,m2; s:={e=0.0934,sin(omega*t)=alpha};
  s1:={op(s),cos(omega*t)=(1-alpha^2)^(1/2)};
  s2:={op(s),cos(omega*t)=-(1-alpha^2)^(1/2)};
  arc:=180*60/evalf(Pi);
  m1:=maximize(subs(s1,expand(err*arc)),alpha=-1..1);
  m2:=maximize(subs(s2,expand(err*arc)),alpha=-1..1);
  max(m1,m2); end proc:
> MARS:={a=227.9,e=0.0934,omega=2*Pi/1.881}:
  Mars:=plot([subs(MARS,a*(1-e^2)/(1+e*cos(t))),t,t=0..2*Pi],sc
  aling=CONSTRAINED,coords=polar,color=black,legend="Kepler",th
  ickness=2):
```

Ellipse Keplerienne

Equation polaire de l'ellipse. Les constantes du mouvement sont a (demi-grand axe) et e (excentricité).On va chercher θ en fonction de t.

```
> r:=a*(1-e^2)/(1+e*cos(theta(t)));
```

Calcul de la constante de la loi des aires à partir de l'aire de l'ellipse, en fonction de la fréquence ω du mouvement.

```
> A:=pi*a^2*(1-e^2)^(1/2); sigma:=A*omega/pi;
```

$$r := \frac{a\,(1-e^2)}{1+e\cos(\theta(t))}$$

$$A := \pi\,a^2\sqrt{1-e^2}$$

$$\sigma := a^2\sqrt{1-e^2}\;\omega$$

Equation horaire du mouvement Keplerien - Modèle de Copernic

La loi des aires donne une relation entre la dérivée de θ et r, donc une équation différentielle vérifiée par θ.
On écrit l'équation à l'ordre 1 en e.

```
> eq1:=diff(theta(t),t)=sigma/r^2; dleq(eq1);
```

$$eq1 := \frac{d}{dt}\theta(t) = \frac{\omega\,(1+e\cos(\theta(t)))^2}{(1-e^2)^{(3/2)}}$$

$$\left(\frac{d}{dt}\theta(t)\right) = \omega + 2\,\omega\cos(\theta(t))\,e + \mathrm{O}(e^2)$$

Pour "séparer" les variables, on inverse 1+2e.cos(θ), ce qui à l'ordre 1 donne 1-2e.cos(θ).
La nouvelle équation est juste la précédente multipliée par ce "multiplicateur" 1-2e.cos(θ).
On vérifie qu'à l'ordre 1 en e, elle est très simple.

```
> mult:=1-2*e*cos(theta(t));
  eq2:=mult*diff(theta(t),t)=mult*rhs(eq1); dleq(eq2);
```

$$mult := 1 - 2\,e\cos(\theta(t))$$

$$eq2 := (1-2\,e\cos(\theta(t)))\left(\frac{d}{dt}\theta(t)\right) = \frac{(1-2\,e\cos(\theta(t)))\,\omega\,(1+e\cos(\theta(t)))^2}{(1-e^2)^{(3/2)}}$$

$$\left(\frac{d}{dt}\theta(t)\right) - 2\cos(\theta(t))\left(\frac{d}{dt}\theta(t)\right)e = \omega + \mathrm{O}(e^2)$$

On développe θ par raport à e. L'équation précédente peut s'interpréter comme un système en les coefficients du dévellopement.
Il ne reste qu'à résoudre ces deux équations très simples :

```
> subs(theta(t)=sum(Theta[i](t)*e^i,i=0..1),eq2): dleq(%);
  EQ_HORAIRE:=dlsolve(eq2,theta,2);
```

$$\left(\frac{d}{dt}\Theta_0(t)\right) + \left(\left(\frac{d}{dt}\Theta_1(t)\right) - 2\cos(\Theta_0(t))\left(\frac{d}{dt}\Theta_0(t)\right)\right)e + \mathrm{O}(e^2) = \omega + \mathrm{O}(e^2)$$

$$EQ\_HORAIRE := \omega\,t + 2\sin(\omega\,t)\,e$$

On trouve alors les coordonnées approchées du point à l'ordre 1 : on connaît θ en fonction de t, et r en fonction de θ donc de t.
En prenant le développement à l'ordre 1 on trouve, et ceci prouve que Copernic savait calculer !, le modèle utilisé par Copernic et Rhéticus.

```
> COORDONNEES:=subs(theta(t)=EQ_HORAIRE,[r*cos(theta(t)),r*sin(
  theta(t))]); COPERNIC:=dl2(COORDONNEES);
```

$$COORDONNEES :=$$

$$\left[\frac{a\,(1-e^2)\cos(\omega\,t+2\sin(\omega\,t)\,e)}{1+e\cos(\omega\,t+2\sin(\omega\,t)\,e)},\ \frac{a\,(1-e^2)\sin(\omega\,t+2\sin(\omega\,t)\,e)}{1+e\cos(\omega\,t+2\sin(\omega\,t)\,e)}\right]$$

$$COPERNIC := \left[a\cos(\omega\,t) - \frac{3\,a\,e}{2} + \frac{1}{2}\,a\,e\cos(2\,\omega\,t),\ a\sin(\omega\,t) + \frac{1}{2}\,a\,e\sin(2\,\omega\,t)\right]$$

```
> COP_plot:=plot([op(subs(MARS,COPERNIC)),t=0..1.881],color=green,legend="Copernic",scaling=CONSTRAINED,thickness=2):
display(COP_plot,Mars);
```



Il ne reste plus qu'à calculer la valeur de θ avec ce modèle et à le comparer à la valeur théorique.
Evidemment il y aura parfaite adéquation à l'ordre 1. Les problèmes sont donc à partir de l'ordre 2 !

```
> copernic:=dltheta(COPERNIC,3); theta2:=dlsolve(eq2,theta,3);
erreur_c2:=dlcmp(copernic,theta2,2);
```

$$copernic := -\ln(\cos(\omega t) + \sin(\omega t)\, I)\, I + 2\sin(\omega t)\, e + e^2 \sin(2\,\omega t)$$

$$\theta 2 := \omega t + 2\sin(\omega t)\, e + \frac{5}{4} e^2 \sin(2\,\omega t)$$

$$erreur\_c2 := -\frac{1}{4} e^2 \sin(2\,\omega t)$$

On peut donc évaluer les erreurs faites par Copernic (en minutes d'arc, avec 1'=(π/180)/60= .2908882087e-3.).
L'excentricité de l'orbite de Mars est environ 0.0934.

```
> arcmax(erreur_c2);
```

$$7.497347555$$

Pour mémoire, voici les calculs à l'ordre 3.

```
> copernic3:=dltheta(COPERNIC,4); theta3:=dlsolve(eq2,theta,4);
erreur_c3:=dlcmp(copernic3,theta3,3);
```

$copernic3 :=$

$$-\ln(\cos(\omega t) + \sin(\omega t)\, I)\, I + 2\sin(\omega t)\, e + e^2 \sin(2\,\omega t) - \frac{3}{2} e^3 \sin(\omega t) + \frac{7}{6} e^3 \sin(3\,\omega t)$$

$$\theta 3 := \omega t + 2\sin(\omega t)\, e + \frac{5}{4} e^2 \sin(2\,\omega t) + \frac{13}{12} e^3 \sin(3\,\omega t) - \frac{1}{4} e^3 \sin(\omega t)$$

$$erreur\_c3 := -\frac{1}{4} e^2 \sin(2\,\omega t) + \left(-\frac{5}{4}\sin(\omega t) + \frac{1}{12}\sin(3\,\omega t)\right) e^3$$

```
> arcmax(erreur_c3);
```

$$9.949887886$$

Modèle vicariant

```
> SE:=[-2*e*a,0]; EM:=[rho*cos(omega*t),rho*sin(omega*t)];
SM:=SE+EM;
```

$$SE := [-2\, a\, e, 0]$$
$$EM := [\rho \cos(\omega t), \rho \sin(\omega t)]$$
$$SM := [\rho \cos(\omega t) - 2\, a\, e, \rho \sin(\omega t)]$$

```
> CM:=a; EC:=3*e*a/4; rho2:=EC^2+CM^2+2*EC*CM*cos(C);
```

$$CM := a$$
$$EC := \frac{3\, a\, e}{4}$$
$$\rho 2 := \frac{9\, a^2 e^2}{16} + a^2 + \frac{3}{2} a^2 e \cos(-\omega t + M)$$

```
> sinM:=3*e*sin(omega*t)/4; cosM:=(1-sinM^2)^(1/2);
C:=omega*t-M;
```

$$sinM := \frac{3}{4}\sin(\omega t)\, e$$
$$cosM := \sqrt{1 - \frac{9}{16}\sin(\omega t)^2 e^2}$$
$$C := \omega t - M$$

```
> subs({sin(M)=sinM,cos(M)=cosM},expand(rho2));
rho1:=(%)^(1/2);
```

$$\frac{9\, a^2 e^2}{16} + a^2 + \frac{3}{2} a^2 e \cos(\omega t)\sqrt{1 - \frac{9}{16}\sin(\omega t)^2 e^2} + \frac{9}{8} a^2 e^2 \sin(\omega t)^2$$

$$\rho 1 := \sqrt{\frac{9\,a^2\,e^2}{16} + a^2 + \frac{3}{2}\,a^2\,e\cos(\omega\,t)\,\sqrt{1 - \frac{9}{16}\sin(\omega\,t)^2\,e^2} + \frac{9}{8}\,a^2\,e^2\sin(\omega\,t)^2}$$

```
> VICARIANT:=combine(expand(simplify(dl2(subs(rho=rho1,SM)),symbolic)),trig);
```

$$VICARIANT := \left[a\cos(\omega\,t) + \frac{3}{8}\,a\,e\cos(2\,\omega\,t) - \frac{13\,a\,e}{8},\; a\sin(\omega\,t) + \frac{3}{8}\,a\,e\sin(2\,\omega\,t)\right]$$

```
> VIC_plot:=plot([op(subs(MARS,VICARIANT)),t=0..1.881],color=red,legend="Vicariant",scaling=CONSTRAINED,thickness=2):
display(VIC_plot,Mars);
```




```
> display(COP_plot,VIC_plot);
```




```
> vicariant:=dltheta(VICARIANT,3); theta2;
erreur_v2:=dlcmp(vicariant,theta2,2);
```

$$vicariant := -\ln(\cos(\omega\,t) + \sin(\omega\,t)\,I)\,I + 2\sin(\omega\,t)\,e + \frac{5}{4}\,e^2\sin(2\,\omega\,t)$$

$$\omega\,t + 2\sin(\omega\,t)\,e + \frac{5}{4}\,e^2\sin(2\,\omega\,t)$$

$$erreur\_v2 := 0$$

```
> vicariant:=dltheta(VICARIANT,4);
erreur_v3:=dlcmp(vicariant,theta3,3);
```

$$vicariant := -\ln(\cos(\omega\,t) + \sin(\omega\,t)\,I)\,I + 2\sin(\omega\,t)\,e + \frac{5}{4}\,e^2\sin(2\,\omega\,t) - \frac{39}{32}\,e^3\sin(\omega\,t) + \frac{139}{96}\,e^3\sin(3\,\omega\,t)$$

$$erreur\_v3 := \left(-\frac{31}{32}\sin(\omega\,t) + \frac{35}{96}\sin(3\,\omega\,t)\right)e^3$$

```
> arcmax(erreur_v3);
```

3.734678728

```
>
```